// File: @openzeppelin/contracts/utils/Context.sol


// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol


// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/math/Math.sol


// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: contracts/StringBuilderLib.sol


pragma solidity ^0.8.4;


library StringBuilderLib {
	bytes16 private constant _SYMBOLS = "0123456789";

    struct StringBuilder {
		bytes buf;
		uint256 len;
	}

	function newStringBuilder(uint256 n) internal pure returns (StringBuilder memory) {
		return StringBuilder({
			buf: new bytes(n),
			len: 0
		});
	}

	function writeString(
		StringBuilder memory stringBuilder, 
		string memory s
	) internal pure {
		writeBytes(stringBuilder, bytes(s));
	}

	function writeBytes(
		StringBuilder memory stringBuilder, 
		bytes memory b
	) internal pure {
		uint256 len = stringBuilder.len;
		for (uint256 i = 0; i < b.length; i++) {
			stringBuilder.buf[len + i] = b[i];
		}
		stringBuilder.len += b.length;
	}

	function writeChar(
		StringBuilder memory stringBuilder, 
		string memory s
	) internal pure {
		stringBuilder.buf[stringBuilder.len] = bytes(s)[0];
		stringBuilder.len++;
	}

	// From @openzeppelin/contracts/utils/Strings.sol, modified to include a fixed point & write to stream.
	function writeFixed(
		StringBuilder memory stringBuilder, 
		uint256 value
	) internal pure {
		unchecked {
			bytes memory buf = stringBuilder.buf;
			uint256 len = stringBuilder.len;

			// @dev Add an additonal 1 for the fixed point
            uint256 length = Math.log10(value) + 1 + 1; 
			uint256 i = 0;
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buf, add(add(len, 32), length))
            }
            while (true) {
				// @dev include the fixed point to account for div 100000
				if (i == 1) {
					ptr--;
					/// @solidity memory-safe-assembly
                	assembly {
                    	mstore8(ptr, byte(0, "."))
                	}
				}
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
				i += 1;
            }
			stringBuilder.len += length;
			// @dev loop back and decrement length to trim trailing zeros
			while (true) {
				bytes1 char = stringBuilder.buf[stringBuilder.len - 1];
				if (char == ".") {
					stringBuilder.len--;
					break;
				}
				if (char != "0") {
					break;
				}
				stringBuilder.len--;
			}
        }
	}

	function trimOne(StringBuilder memory stringBuilder) internal pure {
		stringBuilder.len--;
	}

	function toBytes(
		StringBuilder memory stringBuilder
	) internal pure returns (bytes memory) {
		bytes memory buf = stringBuilder.buf;
		uint256 len = stringBuilder.len;
        /// @solidity memory-safe-assembly
		assembly {
			mstore(buf, len)
		}
		return buf;
	}

	function toString(
		StringBuilder memory stringBuilder
	) internal pure returns (string memory) {
		return string(toBytes(stringBuilder));
	}
}
// File: contracts/PathLib.sol


pragma solidity ^0.8.4;


library PathLib {
    function decodePath(
		StringBuilderLib.StringBuilder memory stringBuilder, 
		bytes memory path
	) internal pure {
        uint256 i = 0;
        while (i < path.length) {
            uint8 u = uint8(path[i]);
            if (u == 128) {
				StringBuilderLib.writeChar(stringBuilder, "M");
                i++;
            } else if (u == 129) {
				StringBuilderLib.writeChar(stringBuilder, "C");
                i++;
            } else if (u == 130) {
				StringBuilderLib.writeChar(stringBuilder, "Z");
                i++;
            } else {
				StringBuilderLib.writeFixed(stringBuilder, readUint16(path, i));
				i += 2;

				StringBuilderLib.writeChar(stringBuilder, ",");
				
				StringBuilderLib.writeFixed(stringBuilder, readUint16(path, i));
                i += 2;
            }
			StringBuilderLib.writeChar(stringBuilder, " ");
        }
		StringBuilderLib.trimOne(stringBuilder);
    }

    function readUint16(bytes memory b, uint start) private pure returns (uint16) {
        uint16 x;
        assembly {
            x := mload(add(b, add(0x02, start)))
        }
        return x;
    }
}
// File: @openzeppelin/contracts/utils/Strings.sol


// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;


/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: @openzeppelin/contracts/utils/Base64.sol


// OpenZeppelin Contracts (last updated v4.7.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        /// @solidity memory-safe-assembly
        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: contracts/MetadataGenerator.sol




pragma solidity ^0.8.4;







contract MetadataGenerator is Ownable {
	struct Color {
		string name;
		string value;
	}

    struct Body {
        string name;
        bytes path;
		uint256 destlen;
    }

	struct Face {
        string name;
        bytes path;
		uint256 destlen;
    }

    mapping(uint8 => Color) public colors;
    mapping(uint8 => Body) public bodies;
    mapping(uint8 => Face) public faces;
	
	string public externalUrlPrefix;

	constructor(string memory _externalUrlPrefix) {
		externalUrlPrefix = _externalUrlPrefix;
	}

	function setColors(
		uint8[] calldata ids,
		string[] calldata names,
		string[] calldata values
	) external onlyOwner {
		for (uint256 i = 0; i < ids.length; i++) {
			colors[ids[i]] = Color({
				name: names[i], 
				value: values[i]
			});
		}
	}

    function setBodies(
		uint8[] calldata ids,
		string[] calldata names,
		bytes[] calldata paths,
		uint256[] calldata destlens
	) external onlyOwner {
		for (uint i = 0; i < ids.length; i++) {
			bodies[ids[i]] = Body({ 
				name: names[i], 
				path: paths[i],
				destlen: destlens[i]
			});
		}
    }

    function setFaces(
		uint8[] calldata ids,
		string[] calldata names,
		bytes[] calldata paths,
		uint256[] calldata destlens
	) external onlyOwner {
		for (uint i = 0; i < ids.length; i++) {
			faces[ids[i]] = Face({ 
				name: names[i], 
				path: paths[i],
				destlen: destlens[i]
			});
		}
    }

	function setExternalUrlPrefix(string memory _externalUrlPrefix) external onlyOwner {
		externalUrlPrefix = _externalUrlPrefix;
	}

	function generateMetadata(uint256 tokenId) external view returns (string memory) {
		(uint8 colorId, uint8 bodyId, uint8 faceId) = splitTokenId(tokenId);
		
		Color memory color = colors[colorId];
		Body memory body = bodies[bodyId];
		Face memory face = faces[faceId];

		// 341 is the size of the other text for the svg.
		uint256 buffSize = 341 + body.destlen + face.destlen;
		StringBuilderLib.StringBuilder memory stringBuilder = StringBuilderLib.newStringBuilder(buffSize);
		generateSvg(stringBuilder, color.value, body.path, face.path);
		bytes memory svg = StringBuilderLib.toBytes(stringBuilder);

		string memory tokenIdString = Strings.toString(tokenId);
		
		return string(abi.encodePacked("data:application/json;base64,", Base64.encode(abi.encodePacked(
			"{\"image\":\"data:image/svg+xml;base64,", Base64.encode(svg),
			"\",\"name\":\"Buddy #", tokenIdString,
			"\",\"external_url\":\"", externalUrlPrefix, tokenIdString,
			"\",\"attributes\":[{\"trait_type\":\"Color\",\"value\":\"", color.name, 
			"\"},{\"trait_type\":\"Body\",\"value\":\"", body.name, 
			"\"},{\"trait_type\":\"Face\",\"value\":\"", face.name, 
			"\"}]}"
		))));
	}

	function splitTokenId(uint256 tokenId) private pure returns (uint8, uint8, uint8) {
		require(tokenId < 1000);
		uint256 color = tokenId / 100;
		tokenId = tokenId - (color * 100);
		uint256 body = tokenId / 10;
		uint256 face = tokenId - (body * 10);
		return (uint8(color), uint8(body), uint8(face));
	}

	function generateSvg(
		StringBuilderLib.StringBuilder memory stringBuilder,
		string memory _color,
		bytes memory _body,
		bytes memory _face
	) private pure {
		StringBuilderLib.writeString(stringBuilder, "<?xml version=\"1.0\" encoding=\"UTF-8\" standalone=\"no\"?><svg width=\"1e3\" height=\"1e3\" viewBox=\"0 0 1e3 1e3\" version=\"1.1\" id=\"svg115\" xml:space=\"preserve\" xmlns=\"http://www.w3.org/2000/svg\" xmlns:svg=\"http://www.w3.org/2000/svg\">");
       	writeSvgPath(stringBuilder, "body", _color, _body);
       	writeSvgPath(stringBuilder, "face", "000000", _face);
       	StringBuilderLib.writeString(stringBuilder, "</svg>");
	}

	function writeSvgPath(
		StringBuilderLib.StringBuilder memory stringBuilder,
		string memory id,
		string memory color,
		bytes memory path
	) private pure {
		StringBuilderLib.writeString(stringBuilder, "<path id=\"");
		StringBuilderLib.writeString(stringBuilder, id);
		StringBuilderLib.writeString(stringBuilder, "\" d=\"");
		PathLib.decodePath(stringBuilder, path);
		StringBuilderLib.writeString(stringBuilder, "\" fill=\"#");
		StringBuilderLib.writeString(stringBuilder, color);
		StringBuilderLib.writeString(stringBuilder, "\" stroke=\"#");
		StringBuilderLib.writeString(stringBuilder, color);
		StringBuilderLib.writeString(stringBuilder, "\" />");
	}
}